RISC-Linz
Research Institute for Symbolic Computation
Johannes Kepler University
A-4040 Linz, Austria, Europe

Algorithms for Indefinite Summation of Rational Functions in Maple

Roberto PIRASTU

(February 9 , 1995)

RISC-Linz Report Series No. 95-09

Supported by: Commission of the European Communities in the framework of the program "Human Capital and Mobility", contract nr. ERBCHBICT930501.

Published in: MapleTech, The Maple Technical Newsletter, Vol. 2, N. 1, (1995).

Algorithms for Indefinite Summation of Rational Functions in Maple

Roberto Pirastu[1]

Introduction

In many parts of mathematics and computer science sum expressions like $g(n) := \sum_{k=1}^{n} f(k)$ arise in a natural way, for instance in combinatorics or complexity analysis. Usually one is interested in finding a *closed form* for such an expression as an expression in n, or at least in studying its asymptotic behaviour as $n \to \infty$ by computing some suitable representation.

For instance, during basic calculus courses every student has to compute the value of expressions like

$$\sum_{k=1}^{\infty} \frac{5}{k(k+2)} = \frac{5}{1 \cdot 3} + \frac{5}{2 \cdot 4} + \frac{5}{3 \cdot 5} + \cdots$$

This can be easily done by Maple with the help of the function `sum`.

```
> sum(5/(k*(k+2)), k=1..infinity);
```

$$\frac{15}{4}$$

The way Maple obtains this result is first to represent the sum $g(n) = \sum_{k=1}^{n} \frac{5}{k(k+2)}$ in closed form, i.e., as a rational function in the variable n, and then to compute the value of the limit of $g(n)$ as $n \to \infty$. From the observation that

$$\frac{5}{k(k+2)} = h(k+1) - h(k) \qquad \text{where} \qquad h(k) = -\frac{5}{2}\frac{2k+1}{k(k+1)} \tag{1}$$

we obtain the following rational representation of $g(n)$

$$g(n) = \sum_{k=1}^{n} \frac{5}{k(k+2)} = \sum_{k=1}^{n} (h(k+1) - h(k)) = h(n+1) - h(1) = \frac{5}{4}\frac{n(3n+5)}{(n+1)(n+2)} \tag{2}$$

On the other hand, such a rational representation does not always exist. As an example, the harmonic numbers $H_n = \sum_{k=1}^{n} \frac{1}{k}$ are not expressible as a rational function in n, as we will see later.

As a second example, consider the well known quicksort algorithm and denote by $F(n)$ the average number of pairwise comparisons needed to sort n symbols. After some observations one obtains the following formula for $F(n)$

$$F(n) = (n+1) \sum_{k=1}^{n-1} \frac{2k}{(k+1)(k+2)}$$

[1]RISC-Linz, J. Kepler University, Linz, Austria. Roberto.Pirastu@risc.uni-linz.ac.at
Supported by the Commission of the European Communities in the framework of the program "Human Capital and Mobility", contract Nr. ERBCHBICT930501

and one may then ask for a closed form for $F(n)$. Maple computes the following decomposition as in equation (1)

```
> sum( 2*k/(k+1)/(k+2), k);
```

$$\frac{2}{k+1} + 2\,\Psi(k+2)$$

The Ψ function is the logarithmic derivative of the Γ function, viz. $\Psi(x) = \frac{d}{dx}\log\Gamma(x)$. From the basic property $\Gamma(x+1) = x\Gamma(x)$ it follows that $\Psi(x+1) - \Psi(x) = 1/x$. The Maple output corresponds to the following rewriting of the summand in a summable part and a remainder, which cannot be further reduced in the same way using rational functions

$$\frac{2k}{(k+1)(k+2)} = \frac{2}{k+2} - \frac{2}{k+1} + \frac{2}{k+2}$$

or, equivalently, the decomposition

$$\sum_{k=0}^{n-1} \frac{2k}{(k+1)(k+2)} = -\frac{2n}{n+1} + \sum_{k=0}^{n-1} \frac{2}{k+2} = -\frac{2n}{n+1} + 2\left(H_n - 1 + \frac{1}{n+1}\right) = 2(n+1)H_n - 4n$$

In this case a rational expression for the sum does not exist, but, nevertheless, the decomposition above is useful for asymptotic analysis. The last expression completes the analysis, because it is a well-known fact that the asymptotic behaviour of H_n is given by $H_n = \log n + \gamma + O(\frac{1}{n})$, where γ denotes the Euler constant ($\gamma = 0.57721...$).

This motivates the following formulation of the problem of *indefinite summation of rational functions* (IRS): for a given proper rational function $f(x)$, compute rational functions $h(x)$ and $r(x)$ such that

$$f(x) = h(x+1) - h(x) + r(x) \tag{3}$$

In addition, as a solution with $r(x) = 0$ corresponds to a *rational closed form* for sums over $f(x)$ like in (2), we are interested in keeping the non summable remainder $r(x)$ *as small as possible*. As we shall see below minimality will be defined with respect to the degree of the denominator polynomial.

We should remark that in the (IRS) problem of indefinite summation, the summand $f(x)$ does not depend on the bounds of summation, i.e., it is independent of n. Otherwise one speaks of the different problem of *definite summation*. Several aspects of definite summation in connection with the use of Maple, in particular for proving combinatorial identities, are treated by Strehl in [10]. In this paper mainly hypergeometric summands are considered, whose indefinite summation can be treated, for example, by Gosper's algorithm (see [2]). For a survey on the Maple facilities for solving indefinite summation see [4].

In this article we present three different algorithms for the problem of indefinite rational summation. The implementation available in Maple is based on work of Moenck, published in 1977 (see [3]). It was noticed by Paule in 1992, that the Maple function `sum` does not always lead to a correct answer. We will point out later that this failure is not due to an incorrect implementation in Maple, but to several gaps in the description given by Moenck. The implementation has now been corrected (see [6]). Abramov proposed as early as 1975 a completely different approach for solving the problem (IRS), which turns out to admit an efficient implementation. We propose a slight modification which shows a better behaviour with respect to the size of the result, as we describe later. An extensive theoretical treatment of the subject was provided by Paule in 1992 (see [5]), where he suggests an algorithm analogous to the Horowitz method for integration of rational functions. It is the goal of this article to describe and compare briefly these algorithms which have been implemented by the present author in the Maple system. All procedures are available as a package named `ratsum` under anonymous ftp at `ftp.risc.uni-linz.ac.at` in the directory `combinatorics/maple`.

Both the algorithm of Abramov and of Paule are structured in such a way that they suggest the use of parallel architectures. In [8] an efficient implementation is presented, which works on a distributed network in ||MAPLE||, a parallel computer algebra system based on Maple and the language Strand.

Indefinite Summation

The algorithms we present here compute a decomposition $h(x), r(x)$ as in (3), where the denominator of $r(x)$ has minimal degree. Following Abramov we call $r(x)$ a *bound* for $f(x)$. Later we will also discuss the fact that the bound of a rational function is not uniquely determined.

In order to explain the methods, we have to look at the structure of the denominator of the rational function. Let the summand $f(x) = p(x)/q(x)$ be a rational function over some coefficient field (here we consider mainly the field of rational numbers), given by two relatively prime polynomials $p(x)$ and $q(x)$. As an example let

$$f(x) = \frac{p(x)}{q(x)} = \frac{x^2+1}{x^7+23x^6+218x^5+1090x^4+3033x^3+4455x^2+2700x}$$

Maple computes the factorization of $q(x)$ as follows

```
> q:=x^7+23*x^6+218*x^5+1090*x^4+3033*x^3+4455*x^2+2700*x:
> factor(q);
```

$$x\,(x+4)\,(x+5)^2\,(x+3)^3$$

From the factorization we get the so-called *shift structure* of the polynomial. Let us collect all irreducible factors $q_1(x), q_2(x), \ldots, q_l(x)$ of $q(x)$ which are shift equivalent, i.e., for all $k \leq l$ $q_1(x) = q_k(x+i)$ for some integer i. In our example, then, all factors are in the same class. So, choose $q_1(x) = x$, then we have

$$q(x) = q_1(x)\,(q_1(x+3))^3\,q_1(x+4)\,(q_1(x+5))^2$$

We represent this situation graphically as in Fig. 1: We put m boxes on the i-th place on a line when $q_1(x+i)$ has multiplicity m as a factor of $q(x)$.

The maximal distance of stacks of boxes in this graph, i.e., the maximal integer distance between roots of $q(x)$, plays a fundamental role in the problem; it is called the *dispersion* of the polynomial (see [1]). In our example the dispersion of $q(x)$ is five. If the polynomial splits in more than one shift component, then the dispersion is the maximum among the dispersions of all the components. For instance, one can easily see that if the sum admits a rational closed form, then the denominator of the summand has non-zero dispersion. This provides a first criterion for the non-existence of a rational closed form for the sum of a given rational function. In particular, it follows that harmonic numbers do not admit a rational closed form. Here the summand $1/x$ has dispersion zero, i.e., the graph of the shift structure consists in one single box.



Figure 1: Shift-Structure of $x(x+3)^3(x+4)(x+5)^2$

The computation of the dispersion can be carried out essentially in two different ways. The dispersion of a polynomial $q(x)$ is the greatest integer d such that $q(x)$ and $q(x+d)$ have a non-trivial common divisor. This means that we can obtain d as the maximal integer root of the resultant $Res(q(x), q(x+d))$ with respect to x (the resultant then is a polynomial in d). The following short Maple procedure does the computation.

```
> dis := proc( pol, var)
>     map(x -> op( 1, x), roots(resultant( pol, subs( var = var + d, pol), var)));
>     max(op(select(type,",integer))); end:
> dis( q, x);
```

$$5$$

This approach is inherently very time consuming, since the degree of the resultant grows quadraticly with respect to the degree of the original polynomial. If efficient factorization methods are available, for example when working over the field of rational numbers, then the shift structure can be computed from the fully factored form of the polynomial.

It can be shown that for a solution $h(x), r(x)$ for (3) the denominator of $r(x)$ has minimal degree if, and only if the dispersion of the denominator is zero (see [1], for a proof see [5, 6]). All three algorithms implicitly make use of this fact in the computations.

Moenck's Algorithm

Consider the rational function

$$f(x) = \frac{p(x)}{q(x)} = \frac{x^2+3}{x^2(x+2)^3(x+3)^2(x^2+2)(x^2+2x+3)^2}$$

The shift structure of the denominator $q(x)$ gives a decomposition into two classes, represented in the left part of Fig. 2.

The first step "fills up the gaps" in each class, i.e., we put as many boxes on each line as we need to get a rectangle on it. This means that we have to take all factors with the maximal multiplicity arising in that class. This way we obtain the new denominator as in the right part of Fig. 2. This brings along a new representation for the rational function as follows.

$$f(x) = \frac{x(x+1)^3(x+3)(x^2+2)(x^2+3)}{x^3(x+1)^3(x+2)^3(x+3)^3(x^2+2)^2(x^2+2x+3)^2}$$

If we consider the two classes separately, i.e.,

$$q_1(x) = x^3(x+1)^3(x+2)^3(x+3)^3 \qquad \text{and} \qquad q_2(x) = (x^2+2)^2(x^2+2x+3)^2$$



Figure 2: Shift-structure and saturation of $x^2(x+2)^3(x+3)^2(x^2+2)(x^2+2x+3)^2$

then we are able to find polynomials $p_1(x), p_2(x)$ such, that $f(x) = \frac{p_1(x)}{q_1(x)} + \frac{p_2(x)}{q_2(x)}$, because $q_1(x)$ and $q_2(x)$ are relatively prime.

It can be shown that it is sufficient to find solutions to the problem for each shift component separately. In other words, if in our example we find a solution $h_1(x), r_1(x)$ for $\frac{p_1(x)}{q_1(x)}$ and $h_2(x), r_2(x)$ for $\frac{p_2(x)}{q_2(x)}$, then $h(x) = h_1(x) + h_2(x)$ and $r(x) = r_1(x) + r_2(x)$ are a solution for $f(x)$. Analogously for more than two shift components.

Consider now the decomposition $f(x) = \frac{p_1(x)}{q_1(x)} + \frac{p_2(x)}{q_2(x)}$ (for some $p_1(x), p_2(x)$). First, we go on computing a complete decomposition with respect to the first class

$$\frac{p_1(x)}{q_1(x)} = \frac{p_1(x)}{x^3(x+1)^3(x+2)^3(x+3)^3} = \sum_{j=0}^{3} \frac{p_{1,j}(x)}{(x+j)^3(x+j+1)^3 \cdots (x+3)^3} \tag{4}$$

where $\deg p_{1,j}(x) < \deg(x+3)^3 = 3$. We iteratively decompose each summand as in (3) with a remainder having smaller dispersion after each step.

Consider now any summand of the right-hand side of equation (4) and compute a polynomial solution $\alpha(x), \beta(x)$ to the equation

$$(x+j)^3\alpha(x) + ((x+3)^3 - (x+j)^3)\beta(x) = p_{1,j}(x)$$

This can be done by `gcdex`, since $(x+j)^3$ and $((x+3)^3 - (x+j)^3)$ are relatively prime. Then one can verify that

$$\frac{p_{1,j}(x)}{(x+j)^3 \cdots (x+3)^3} = \frac{-\beta(x+1)}{(x+j+1)^3 \cdots (x+3)^3} - \frac{-\beta(x)}{(x+j)^3 \cdots (x+3)^3} + \frac{\beta(x+1) - \beta(x) + \alpha(x)}{(x+j+1)^3 \cdots (x+3)^3}$$

and one obtains a decomposition as in (3) for the j-th summand. We iterate the procedure till we obtain a new remainder with dispersion zero (or zero itself), then we sum up all sub-results finding a decomposition as in (3) for $f(x)$.

This method can be seen as discrete analogue to Hermite's algorithm for the integration of rational functions (see [11]).

The problem in Moenck's method lies in the computation of the shift saturation of the denominator. In his work the definition of such a *shift saturation* does not make sure that the classes are relatively prime. Using the `printlevel` facility one sees that in Maple the denominator is decomposed into three classes $q_1(x) = x^2(x+2)^2(x+3)^2$, $q_2(x) = x+2$ and $q_3(x) = (x^2+2)^2(x^2+2x+3)^2$. From this follows that, in this case, the partial fraction decomposition of $f(x)$ can not be computed. To see the problem, let us try to get a solution by the function `sum`:

```
> sum ( (x^2+3)/(x^2*(x+2)^3*(x+3)^2*(x^2+2)*(x^2+2*x+3)^2),x);
Error, (in gcdex/diophant) wrong number (or type) of arguments
```

Furthermore, Moenck does not give any algorithm to fill up the gaps in the shift structure of the denominator. On the other hand, we know the factorization of the polynomial from the computation of the dispersion. This makes the computation of the shift saturation immediate and the rest of the method can then be applied to the correct decomposition.

Paule's Algorithm

In [5] Paule provides a general algebraic frame (*greatest factorial factorization*) for rational and hypergeometric summations and presents interesting general results on the structure of the set of solutions of (3). Analogously to Horowitz's algorithm for integration of rational functions he reduces the problem to solving a system of linear equations. Since this algorithm is also based on the Hermite-Ostrogradsky method, it produces the same result as the corrected version of Moenck.

Let us consider the following rational function with shift structure as described in the left part of Fig. 3.

$$f(x) = \frac{x^2+1}{x(x+1)^4(x+3)}$$

We again need what Paule calls the *shift saturated extension* (SSE) of the denominator, i.e., each stack of boxes in the class must have the same height. In [5] and [6] algorithms are described, which show how to compute such a saturation without knowing the factorization of the polynomials. However in practice, i.e., working over suitable domains, the use of the factorization facility in Maple is more practical and results in much higher efficiency.

Let us rewrite equation (3) with $h(x) = \gamma(x)/\delta(x)$ and $r(x) = \epsilon(x)/\eta(x)$

$$f(x) = \frac{\alpha(x)}{\beta(x)} = \frac{x^3(x+2)^4(x+3)^3(x^2+1)}{x^4(x+1)^4(x+2)^4(x+3)^4} = \frac{\gamma(x+1)}{\delta(x+1)} - \frac{\gamma(x)}{\delta(x)} + \frac{\epsilon(x)}{\eta(x)} \tag{5}$$

Paule proved that this equation can be solved using the Ansatz $\delta(x) = \gcd(\beta(x), \beta(x-1)) = x^4(x+1)^4(x+2)^4$ and $\eta(x) = \beta(x)/\delta(x) = (x+3)^4$. In other words, one takes as the denominator for the non summable remainder only the rightmost boxes in each class of the shift saturation and puts in the full rectangle as the denominator of the rational part. In Fig. 3 we illustrate the shift saturated extension of the denominator of the rational function and the decomposition in summable and non summable part.

From the knowledge of $\eta(x)$ and $\delta(x)$ one immediately extracts estimates for the degrees of $\gamma(x)$ and $\epsilon(x)$. So, we substitute these polynomials in indeterminate coefficients into equation (5) obtaining

$$x^3(x+2)^4(x+3)^3(x^2+1) = x^4\gamma(x+1) - (x+3)^4\gamma(x) + \delta(x)\epsilon(x) \tag{6}$$

and, equating the coefficients of same powers of x on the left and on the right-hand side, we get a system of linear equations for the coefficients of $\gamma(x)$ and $\epsilon(x)$.



Figure 3: SSE and decomposition of the denominator

For our example the computation of a solution by Paule's method can be done directly in Maple as follows:

```
> alpha:=x^3*(x+2)^4*(x+3)^3*(x^2+1):
> beta:=x^4*(x+1)^4*(x+2)^4*(x+3)^4:
> delta:=factor(gcd(beta, subs(x=x-1,beta)));
```

$$\delta := x^4\,(x+2)^4\,(x+1)^4$$

```
> gam:=sum( g[i]*x^i, i=0..11): eps:=sum( e[i]*x^i,i=0..3):
> cfs:=solve(identity(alpha = beta/subs(x=x+1,delta)*subs(x=x+1,gam) -
>         (beta/delta)*gam + delta*eps,x), {seq(g[i],i=0..11),seq(e[i],i=0..3)}):
> gam:=factor(subs(cfs,gam));
```

$$gam := -\frac{x^3\,(128+336\,x+912\,x^2+1764\,x^3+2061\,x^4+1491\,x^5+661\,x^6+165\,x^7+18\,x^8)}{24}$$

```
> eps:=factor(subs(cfs,eps));
```

$$eps := -\frac{25}{4} - 4\,x - \frac{3\,x^2}{4}$$

A solution of (3) is then $h(x) = \gamma(x)/\delta(x)$ and $r(x) = \epsilon(x)/\eta(x)$:

$$h(x) = -\frac{128+336x+912x^2+1764x^3+2061x^4+1491x^5+661x^6+165x^7+18x^8}{24x(x+1)^4(x+2)^4}$$

and $r(x) = -(25+16x+3x^2)/4(x+3)^4$. From this example it is also clear that in general the rational solutions $\gamma(x)/\delta(x)$ and $\epsilon(x)/\eta(x)$ are not in reduced form.

Abramov's Algorithm

Let us consider again the example from the last section. We isolate now the rightmost boxes from the rest, viz. $v(x) = (x+3)$ and $w(x) = x(x+1)^4$, and decompose the rational function as

$$f(x) = \frac{x^2+1}{x(x+1)^4(x+3)} = \frac{5x^4+5x^4+15x^2+x+8}{24x(x+1)^4} + \frac{-5}{24(x+3)}$$

Setting $u(x+1) = -5/(24(x+3))$, one can easily see that in the decomposition

$$f(x) = u(x+1) - u(x) + r(x) = -\frac{\frac{5}{24}}{x+3} + \frac{\frac{5}{24}}{x+2} + \frac{-1}{24}\frac{5x^4+5x^3-9x^2-x-16}{x(x+1)^4(x+2)} \qquad (7)$$

the remainder $r(x)$ has dispersion two, while the original function has dispersion three. We now isolate the rightmost boxes in the shift structure of $r(x)$ and iterate the procedure, obtaining at the end a remainder with dispersion zero, i.e., either a bound for $f(x)$ or the zero-element. The latter case corresponds to the fact that the summation over $f(x)$ admits a rational closed form.

Now we are back to dicussing uniqueness of the remainder. Notice that Abramov obtains a remainder with the leftmost boxes in the shift structure, while Moenck and Paule produce a remainder with the rightmost ones. Moreover it can be shown that all possible remainders $r(x)$ for a given summand $f(x)$ are uniquely determined up to integer shift of the factors in the denominator (see [5]).

Let us look more precisely at the shift structure of $r(x)$ after each step. It consists of the boxes in the shift structure of the denominator of $f(x)$, where we erased the rightmost boxes and shifted them one place to the left, as in the right part of Fig. 4. The degree of the denominator in the rational part directly depends on the number of erased boxes.



Figure 4: Shift structure after one step in Abramov's algorithm

In order to take care of the shift structure in a more flexible way, we suggest a variation of this approach. The modification is based on the simple observation that in an analogous way one can reduce the dispersion erasing the leftmost instead of the rightmost boxes of each class, like in the left part of Fig. 4. We now decompose $f(x)$ with respect to the leftmost stack $v(x) = x$ and to the remainder $w(x) = (x+1)^4(x+3)$:

$$f(x) = \frac{x^2+1}{x(x+1)^4(x+3)} = -\frac{x^4+7x^3+18x^2+19x+13}{3(x+1)^4(x+3)} + \frac{1}{3x}$$

If we put $u(x) = -1/3x$ then also in the decomposition

$$f(x) = u(x+1) - u(x) + r(x) = -\frac{1}{3(x+1)} + \frac{1}{3x} - \frac{9x^3+x^2+6x+10}{3(x+3)(x+1)^4}$$

the remainder $r(x)$ has dispersion two and we can iterate again. From this it is clear that we can decide at each step at which endpoints we want to erase/shift a stack of boxes. We propose to make the decision depending on the degree of the denominator of the resulting rational part, i.e., we erase the boxes corresponding to a smaller degree. This is realized by the following short Maple procedure `abrsumnew`.

```
> abrsumnew := proc( f, dis, x)
      local p,q,cp,u,newf;
      if f=0 then RETURN(0) fi;
      if dis = 0 then  RETURN ('Sum'( factor( f), x))  fi;
      p := numer(f); q:=denom(f); cp := gcd(q ,subs( x = x + dis, q));
      if cp = 1 then RETURN( abrsumnew( f, dis-1, x)) fi;
      part (q, cp, 'vp', 'wp'); part (q,subs(x=x-dis,cp),'vm','wm');
      if degree(vm,x)>degree(vp,x) then
          gcdex(vp,wp,p,x,'b','a'); u:=(subs(x=x-1,a/vp));
          newf:=normal(b/wp+u);
      else
          gcdex(vm,wm,p,x,'b','a'); u:=-a/vm;
          newf:=normal(b/wm + subs(x=x+1,a/vm));
      fi;
      u + abrsumnew(newf, dis-1, x);
      end:
```

Here the procedure `part(p,h,'v','w')` computes polynomials $v(x)$, $w(x)$ such, that $v(x)w(x) = p(x)$, $\gcd(v(x), w(x)) = \gcd(w(x), h(x)) = 1$ and $v(x)$ contains only factors arising in $h(x)$.

```
> part := proc ( p, h, v, w)
      local g,v1,w1;
      g:=gcd(p,h);v1:=1;w1:=p;
      while g<>1 do divide(w1,g,'w1'); v1:=v1*g; g:=gcd(w1,v1); od;
      v:=v1;w:=w1;
      end:
```

This modification of Abramov's method often outputs a smaller degree of the denominator of the rational part $h(x)$ than the solutions computed by the other algorithms. In [7] we show that, under certain conditions on the shift structure, we get a minimal result in this sense. Already in our small example we get with the usual Abramov algorithm a rational part with denominator of degree six, while the suggested modification keeps the degree at three. This happens, because the algorithm of Abramov will shift the four boxes at $(x+1)^4$, getting as remainder x^4, while the modification only moves the single boxes corresponding to x, $(x+2)$ and $(x+3)$, getting as remainder $(x+1)^4$.

```
> abrsumnew(f,3,x);
```

$$-\frac{1}{3\,x} - \frac{5}{24\,x+48} - \frac{5}{24\,x+24} + \left(\sum_x \left(-\frac{3\,x^2+4\,x+5}{4\,(x+1)^4}\right)\right)$$

In the output the non summable remainder is expressed an unevaluated sum, without involving the Ψ function. One should remark that in the case of a denominator consisting of several shift classes the algorithm can be applied without modification, in particular without first decomposing the rational function with respect to the shift classes. We just consider the whole dispersion of the polynomial and perform each step on all classes at once.

Comparison

Several tests have been performed on our implementations of the algorithms proposed here, considering different parameters like the number of shift classes, the dispersion of each class, the degree of each factor and its multiplicity. We give a short table of some of the most significant timings obtained (in seconds) on rational functions over the rational numbers.

Shift Classes	Moenck	Paule	d_p	Mod. Abramov	d_a
18		19.03	39	4.93	27
5, 5		74.86	36	19.65	26
2, 5, 9	1606.45	223.20	58	90.63	44
4, 10, 13		118.60	65	36.74	65
4, 10, 13		997.30	92	234.20	75
2, 2, 2, 3, 3, 3		484.33	31	216.65	31
2, 2, 2, 3, 3, 3		2680.30	55	742.85	43

Here the column "Shift Classes" contains the dispersion of each class in the shift structure of the denominator of the input. The column d_p (and d_a resp.) contains the degree of the denominator in the rational part $h(x)$ computed by the algorithm of Paule (and modified Abramov resp.).

As a global observation, the implementation of Abramov's algorithm can be said to be the most efficient one among all three, also with the proposed modification, which needs some more work in the decomposition, though.

The reason for this considerable difference in the behaviour with respect to time can be explained as follows: Applying Abramov's algorithm does not require any saturation of the shift structure of the polynomial but only the value of the dispersion. This means that it deals in general with polynomials of degree not bigger than the original denominator. Let us consider an extremal example like the rational function $f(x) = 1/(x(x+20)^3)$. After the shift saturation operation we get a polynomial of degree 63, i.e., our system blows up to dimension 63. On the other hand, Abramov only needs about 20 gcd-computations with polynomials of degree around three. In fact, for examples of almost shift saturated polynomials the difference of time needed considerably decreases. In these cases also the improvement in the degree of the denominator of the rational part becomes smaller.

Conclusions

Computing closed forms for sum expressions is an important application of symbolic computation to combinatorics. For the case of indefinite summation of rational functions several algorithms are known which admit an efficient implementation in a computer algebra system like Maple.

We briefly described the algorithms of Moenck, Abramov and Paule and the implementations in Maple that we carried out. In particular, the procedure for summation of rational functions in the built-in function `sum` has been repaired. In addition, we proposed a modification of the method of Abramov because it often obtains a result with smaller degree in the denominator of the summable part than the others. All programs are available under anonymous ftp at `ftp.risc.uni-linz.ac.at` in the directory `combinatorics/maple`.

The methods presented here carry along an interesting combinatorial structure, based on the shift-structure of a polynomial, which needs to be studied in some more detail. In any case, this natural decomposition in shift-classes shows an inherent possibility to decompose the problem into considerably smaller sub-problems, which strongly suggests the application of parallel methods. Implementations in ||MAPLE||, a parallel version of Maple (see [9]), have been carried out by K. Siegl and the author (see [8]).

Although the Maple function `sum` already provides many useful algorithmic tools for hypergeometric summations, several extensions are still possible, in particular concerning the efficiency of computations.

References

[1] S. A. Abramov, *The Rational Component of The Solution of a First-Order Linear Recurrence Relation with a Rational Right Side*, Zh. vychisl. Mat. mat. fis. no. 4, **15**, pp.1035-1039 (1975).

[2] R. W. Gosper, *Decision Procedure for Indefinite Hypergeometric Summation.* Proc. Nat. Acad. Sci. U.S.A., **75**, pp.40-42 (1978).

[3] R. Moenck, *On computing closed forms for summations*, Proceedings of MACSYMA users' conference, Berkley, pp.225-236 (1977).

[4] A. Pathria, *Indefinite Summation with Maple*, The Maple Technical Newsletter no. 5, pp.24-31 (1991).

[5] P. Paule, *Greatest Factorial Factorization and Symbolic Summation I*, RISC-Linz Report Series 93-02, J. Kepler University, Linz. *submitted to J. Symb. Comp.* (1993).

[6] R. Pirastu, *Algorithmen zur Summation rationaler Funktionen, (Algorithms for Summation of Rational Functions, in german)*, Diploma Thesis, Univ. Erlangen-Nürnberg (1992).

[7] R. Pirastu, *A Note on the Minimality Problem in Indefinite Summation of Rational Functions*, Actes du Séminaire Lotharingien de Combinatoire, 31 session, Saint-Nabor, J. Zeng, ed., Publications de l'I.R.M.A. 1994/021, pp.95-101 (1993).

[8] R. Pirastu, K. Siegl, *Parallel Computation and Indefinite Summation: a* ||MAPLE|| *Application for the Rational Case*, RISC-Linz Report Series 94-73, J. Kepler University, Linz, to appear in J. of Symb. Comp., special issue *Symbolic Computation in Combinatorics* Δ_1 (1994).

[9] K. Siegl, *Parallelizing Algorithms for Symbolic Computation Using* ||MAPLE||, in Fourth ACM SIGPLAN Symp. on Principles and Practice of Parallel Programming, San Diego, ACM Press, pp.179–186. Also: RISC-Linz Report Series 93-08, J. Kepler University, Linz, (1993)

[10] V. Strehl, *Binomial Sums and Identities*, The Maple Technical Newsletter no. 10, pp.37-49 (1993).

[11] T. N. Subramaniam, D. E. G. Malm, *How to Integrate Rational Functions*, The American Mathematical Monthly no. 8, **99** (1992).